Robert Glaser N3IC
3922 Algiers Road
Randallstown MD 21133

An 8080 Repeater Control System

—part III: software

The finishing touches.

A development system is necessary to write and debug a program of the size and complexity of that of the repeater control. There are commercial development systems available, and hams lucky enough to have access to these systems have the opportunity to modify the program presented here with ease. Medium-to-large-size hobby computers are also equal to the task. I used my personal homemade computer for development. It has 60K of read/write memory, a high level operating system including a text editor and assembler, printer, debugging tools, and the capability to program 2708s. The processor itself is an 8080, so I was able to actually execute the repeater program on it before burning it into ROM.

A good development system is a must when starting from scratch, but if the program is to be used as presented with only code changes, most any hobby computer can be made to program the ROMs. Major modifications would necessitate reassembly.



Fig. 1. Foreground program.

Program Analysis

The repeater control program is fairly long and it may appear quite complicated at first glance. Everything is broken down into manageable subroutines, so it is not too difficult to follow program flow. The software consists of two programs: the foreground program and the interrupt program. The two programs are separate and operate independently. The foreground program counts time, and when it is time for an identificatior it performs the CW ID Touchtones™ interrup the processor, and contrc is passed to the interrup program (which perform whatever task is required The foreground progran may be interrupted at an time, and when the inte rupt service routine exits control returns to th foreground program at th point where it left. This i apparent when listening t the repeater. If the re peater is identifying, and touchtone is sent, the I halts, and, after the tone are handled, the ID re sumes where it left off. Th beauty of the scheme i that the interaction of th programs is handled er tirely by the interrupt harc ware.

Foreground Program

Refer to the program lis ing. At the beginning, som labels are defined. Th various ports are set equa to the proper values CWSPD sets the speed o the CW. At its present set ting, the speed is 19 wpm The CW speed should b proportional to CWSPD IDTM0 through IDTM3 se the time duration betwee successive IDs. This is cur rently set at three minutes

When the 8080 is reset, i begins executing com

mands at address 0. Refer to Fig. 1, a flowchart of the foreground program. At BEGIN, a lock is cleared. The lock permits the removal of the ability to enter the control mode. This will be explained in detail later. Control passes to RESET, where all variables are initialized. All of the output ports are zeroed. A note is in order about how the program handles output. The 8080 can output to its output ports, but it cannot read its output ports back in. Since we need the ability to be able to change only one bit at a time in the output ports, a memory byte is reserved for each output port. Every time the processor outputs data, it writes the output information in the locations OUT0M through OUT7M for ports 0 through 7. This way, if an output bit needs to be changed, the corresponding memory location can be read, the one bit changed, and the byte output. All bits of port 7 are set, because the row and column inputs to the touchtone generator are active low. The stack pointer is loaded, and control jumps around the interrupt location to MASLP.

At MASLP (master loop) the interrupt is enabled, and TIME is checked. If TIME is 0, the system is in the rest mode; as soon as a repeater is used, it will ID. When TIME is 1, the system is counting time to see if it is time to ID. In the program, if TIME is 0, the 150 PTT is checked to see if the repeater is in use. If not, the 450 PTT is checked at MAS2. If neither repeater is in use, the program loops around, continuously waiting for one to be used. When a repeater is activated, either a 150 code or a 450 code is put into MASK. MASK is a variable which tells the CW sending program which repeater to ID. At MAS3, TIME is made 1, and control goes to ID. At ID, TIMER is zeroed. TIMER is a four-byte counter, used to time up to three minutes. The repeater identifies, but before explaining how that occurs, the other path to ID will be explained.

At MASLP, if TIME is 1, control passes to MAS1. In this portion of the loop, the repeater has identified sometime in the past three minutes. In the subsequent three minutes, the processor keeps tab on the repeaters to see which ones should ID later. At MAS1, MASK is modified to reflect which repeaters are in use. TIMER is incremented, and, at MAS6, TIMER is checked to see if it equals IDTM (ID time). If not, three minutes have not elapsed, and the program loops back to MASLP. When time is up, control passes to ID, as before.

At ID, after TIMER is zeroed, MASK is checked to see if either repeater has been utilized in the last three minutes. If not, control resumes at MASLP after clearing TIME, placing the system back into the idle condition. If a repeater has been used, control goes to ID1. At this point, it must be determined which ID message is to be used. IDS (ID status) may have values from 1 to 7. 1 through 5 specify that that ID number is to be used, 6 indicates that the first four should be cycled, and 7 indicates that all five should be cycled. IDN (ID number) specifies the current ID number. IDN goes from 1 to 5. If IDS is between 1 and 5, IDN is set to IDS and control goes to ID3. At ID1, if IDS is 6 or 7, control goes to ID2 where IDN is incremented, advancing to the next ID message. At ID4 and ID5, IDN is checked to see if it is greater than it should be, and if so, it is set back to 1, and control goes to ID3.

At ID3, the HL registers are set to the address of the proper ID message, and the CW sending program is called. After sending the ID, MASK is zeroed and control goes to MASLP.

The CW sending routine is shown in Fig. 2. It is assumed that the address of the message to be sent in CW is in the HL registers, and that MASK indicates which repeaters to send the message to. If the destination is 150, MASK contains C0; if the destination is 450,



Fig. 2. CW routine.



Fig. 3. Interrupt service routine.

MASK contains 30; if the destination is both, MASK contains F0. On entry, the proper transmitters are keyed, keeping them on the air for the duration of the message. At CW0, the character to be sent is fetched. A zero byte indicates that the message is done. If done, the transmitters are unkeyed, and the subroutine returns. Otherwise, at CWND (CW not done), the character is checked to see if it is the special space code of 80. If so, a 6-unit delay is made. A 1-unit delay is appended to every character, so a space is a total of 7 units long. If the character is not the special space code, control goes to CWLET (CW letter). Morse characters are stored left justified, with a 0 representing a dit and a 1 a dah. The byte is shifted left after each dit or dah, and when the byte ends up at 80, the character is done (described in *Byte*, October, 1976, page 36). After CWLET, the tone is turned on. If the character is a dah, an additional delay of 2 units is appended. At CWDOT, the tone is removed, and a trailing 1-unit space is added. The routine loops back to CWLET until the character is finished, where 2 more units are added to create a 3-unit intercharacter delay. At CW2, the next character is fetched and control loops back to CW0. The CW routine is used both by the ID section of the foreground program and various routines in the control section.

The Interrupt Service Program

The interrupt routine is shown in Fig. 3. When the 8080 is interrupted, it goes to address 38. It jumps to TTONE (touchtone), where the service routine is located. Since the foreground program may be interrupted at any time, it is necessary to save all registers. As an error-recovery technique, the stack pointer is checked to see if it is in the limited address space where RAM is located. If not, something is awry, and the program jumps to the beginning, resetting everything. If the stack is okay, MASK is saved, since it may need to be modified by the interrupt programs. OUT1M is saved because some bits are changed there as well. The CW tones are killed, in case an ID has been interrupted (which could leave a constant tone on the repeater until return to the foreground program), and BLK is set high, enabling the blocking function. The decoder is checked to see if the digit is a *, the knockdown digit. If so, the KD output is pulsed for about a millisecond to kill any possible autopatch or remote-base function. If the repeaters are linked, the routine ROGER is called, which sends the "R" in CW. The repeaters are unlinked, and the timeout timer is placed into the timing mode in case a single-digit autopatch was in progress. Control goes to TTON2, the exit point.

If the incoming digit is not a *, LOAD is called, which gets a three-digit code. The code table is checked for the three-digit code. If the code is not found in the table, control goes to TTON2, and nothing happens. If the code is found in the code table, the address of the routine to execute that particular code is obtained. At that point, the program jumps to the particular routine. After the routine is executed, control jumps to



Fig. 4. Load, wait for carrier drop, link, tape, and selective call routines.



Fig. 5. Touchtone test routine.

TTON2.

At TTON2, everything that was saved upon entry of TTONE is restored and the interrupt routine returns to the foreground program.

BITS is a bit set routine used to set a bit in an output byte. The address of the byte is placed in register DE, and a 1 is placed in the desired bit in register B. BITC clears bits the same as BITS sets them.

Shown in Fig. 4, LOAD gets a three-digit code from the touchtone decoder. Upon entry, LOAD waits for SVTT. For user codes, SVTT is immediately present, since it is SVTT which caused the interrupt. For control codes, where several three-digit codes are used, LOAD waits for a code to be entered. When a digit is ready, LOAD calls DECOD. DECOD reads the input ports and decodes the digits into binary form. The digit is stored, and HURRY is called. HURRY checks VTT while counting time. If a tone occurs before three seconds elapse, HURRY returns with the carry clear. If no tone is received in three seconds, HURRY exits with the carry set. The timeout is detected in LOAD, the program is aborted, and LOAD returns. Otherwise DECOD gets the next digit, the sequence repeating. The third digit is fetched in the same manner. After exiting LOAD, either three digits are stored or an invalid code is stored because of failure to send successive digits within three seconds.

DECOD reads the decoder. Presumably, a tone is present when DECOD is called. The digits 1 through 9 are stored as those numbers, and 0, *, and # are stored as decimal 10, 11, and 12. A digit stored as 0 indicates an invalid code. LOAD presets the three digits to 0, so timing out results in one or more stored digits remaining 0

The routine WCD is used to wait for a carrier drop. It is possible to lock out the ROGER routine. If this is done, it also eliminates the need to wait for dropping carrier when controlling the repeater. Upon entry, WCD checks for this, and normally proceeds to check to see if it is in the phone control mode. If so, WCD returns. If not, it checks to see if the control receiver is being used. If so, it waits for the signal there to drop. If not, it waits for the COS signal to disappear. In this manner, WCD only waits when necessary, and waits for the proper signal. The LINK routine checks if the function is to be permitted. If so, it links the repeaters and calls ROGER.

The TAPE routine checks to see if the function is enabled, calls WCD, activates the tape, and exits.

The SELCL (selective call) routine clears BLK, calls WCD, and exits. This permits any tones after 3#3 and before the carrier drop to pass.

TTTST, the touchtone test routine, is shown in Fig. 5. If the function is enabled, GETTT (get touchtone) is called, which loads a sequence of digits. Control goes to TTST1, where the digit count is checked. For each digit, the digit is converted to CW and sent. The addresses of the CW conversions are at DIGAD. The actual CW codes are at CWD1 through CWDP. After the buffer is sent, the digit count is restored and TTTST exits.

The GETTT routine is shown in Fig. 6. Upon entry, the digit count is cleared and register pair DE is initialized to the start of the buffer. If carrier is present at GETT1, the VTT is checked. The program loops until either the carrier is dropped or a digit is received. When the latter happens, DECOD is called and the digit is placed into the buffer. The digit count is incremented, and checked to see if the buffer is full. The buffer is loaded in this manner until the carrier is dropped, when GETTT returns. If the buffer length reaches maximum, WCD is called and then GETTT returns.

When the three-digit control code is sent, the program goes to CNTRL, shown in Fig. 7. If the control mode is locked out, CNTRL exits immediately. Otherwise, WCD is called, and then LOAD. The HL registers are



Fig. 6. Get touchtone routine.



Fig. 7. Control routine.

loaded with the address of the confirm code. Jumping to TTON6 enters TTONE at a point where the code received is checked against the code table, now consisting only of the confirm code. If the received code is not in the single entry table, the interrupt is aborted as usual. If agreement is found, TTONE sends control to CNTR0, a continuation of CNTRL. WCD is called, and CNTRL then loops at CNTR1 until a tone is received. A single-digit code is expected, and DECOD is called to get it. WCD is again called, and if the received digit is invalid, control exits. Otherwise, ROGER is called and the proper program must be selected. If the received digit is between 1 and 7, IDS is loaded with that digit. The command is done, and CNTRL exits. If the digit is 8, CNTRL jumps to IDLD (ID load). A 10, which is digit 0, sends CNTRL to OUT, and 9 has the program jump to RESET, initializing the entire program with the exception of LOCK. If the digit is a *, TIME is cleared; otherwise, the digit must be a # and CNTRL jumps to LNUM (load number). Each routine, at completion, goes to TTON2 and exits.

Fig. 8 shows IDLD. The HL registers are loaded with the address of the programmable ID. The character byte in register B and element count in register C are cleared at IDLD0. IDLD1 waits for a digit to be received, and DECOD is called. If the digit is 3, the stop byte is stored, ROGER is called, and IDLD exits. Otherwise, control goes to IDNTS (ID not stop), where the digit is checked to see if it is a 2. If so, at IDDLT (ID done, left justify) register B is justified by the element count in register C. The character is stored in the message buffer at IDDL (ID done letter), HL is incremented, and control loops to IDLD0. If the digit is not a 2, it is checked to see if it is a 1. If it is, a 1 is shifted into register B and the element bit count is incremented. Otherwise, the digit is checked to see if it is a 0, where a 0 is shifted in. If the digit is not a 3, 2, 1, or a 0, then an invalid digit was sent and it is ignored.

The OUT routine, in Fig. 9, outputs selected bits to the output ports. LOAD is called to get a three-digit code. If the first digit is a *, ROGER is called and OUT exits. Otherwise, the digits are checked to see if they are 0, which is invalid. If an invalid entry is made, after carrier drop, control loops back to OUT. If port 0 is selected, the 10 is changed to a 0 for later use. Several validity checks are made, checking to see if port, bit numbers, and output levels make sense. If they do, ROGER is called. At OUT2, the binary code for the bit number is converted to a 1 in the proper bit of register E. At OPRT (output to port), a machine output instruction is set up in RAM with the required port number. The bit is either set or cleared, and the output instruction in RAM is called. Control loops to OUT, and the cycle continues until OUT is exited with a *.

LNUM (load number) is shown in Fig. 10. The digit count is zeroed, and at LNUM1 LNUM waits for a received digit. DECOD is called, and if the digit is a *, ROGER is called and the routine exits. Otherwise, the digit count is checked and the digit is stored. If more than 11 digits are attempted, the last digit keeps being overwritten.

LOCK has two functions.

```
0001 0000                ;REPEATER CONTROL SYSTEM MONITOR PROGRAM
0002 0000                ;
0003 0000                ;FOR USE WITH AN 8080 CONTROLLER
0004 0000                ;WITH I/O PORTS
0005 0000                ;AND THE NECESSARY EXTERNAL HARDWARE
0006 0000                ;INCLUDING A TOUCH TONE (R) DECODER
0007 0000                ;
0008 0000                ;
0009 0000                ;
0010 0000                ;VERSION 2.0
0011 0000                ;
0012 0000                ;SERIES 2 INCORPORATES ERROR RECOVERY
0013 0000                ;
0014 0000                ;NOVEMBER 1977, ROBERT GLASER N3IC
0015 0000                ;MODIFIED FEBRUARY, 1978
0016 0000                ;COPYRIGHT ROBERT GLASER
0017 0000                ;
0018 0000                ;
0019 0000                ;
0020 0000                PSW:    EQU     6
0021 0000                SP:     EQU     6
0022 0000                PORT1:  EQU     10H
0023 0000                PORT2:  EQU     20H
0024 0000                PORT3:  EQU     30H
0025 0000                PORT4:  EQU     40H
0026 0000                PORT5:  EQU     50H
0027 0000                PORT6:  EQU     60H
0028 0000                PORT7:  EQU     70H
0029 0000                OT5ML:  EQU     0FFH
0030 0000                CWSPD:  EQU     3000
0031 0000                IDTM0:  EQU     0
0032 0000                IDTM1:  EQU     0
0033 0000                IDTM2:  EQU     26
0034 0000                IDTM3:  EQU     0
0035 0000                ;
0036 0000                ;
0037 0000                ;
0038 0000                        ORG     0       ;FIRST RUN
0039 0000                ;INITIALIZATION PROCEDURE
0040 0000                ;
0041 0000                ;
0042 0000                ;
0043 0000 AF             BEGIN:  XRA     A
0044 0001 32 F2 30               STA     LCKR
0045 0004 21 F4 30       RESET:  LXI     H,TIME-1
0046 0007 23             REST1:  INX     H
0047 0008 36 00                  MVI     M,0
0048 000A 3E FF                  MVI     A,OT5ML
0049 000C BD                     CMP     L
0050 000D C2 07 00               JNZ     REST1
0051 0010 AF                     XRA     A
0052 0011 D3 10                  OUT     PORT1
0053 0013 D3 20                  OUT     PORT2
0054 0015 D3 30                  OUT     PORT3
0055 0017 D3 40                  OUT     PORT4
0056 0019 D3 50                  OUT     PORT5
0057 001B D3 60                  OUT     PORT6
0058 001D 3C                     INR     A
0059 001E 32 F3 30               STA     IDS     ;ID STATUS
0060 0021 32 F4 30               STA     IDN     ;ID NUMBER
0061 0024 3E FF                  MVI     A,0FFH
0062 0026 32 FF 30               STA     OUT7M
0063 0029 D3 70                  OUT     PORT7
0064 002B 2A 00 10               LHLD    STCK    ;STACK LOC
0065 002E F9                     SPHL
0066 002F C3 3B 00               JMP     MASLP   ;BYPASS INT LOC
0067 0032                ;
0068 0032                ;
0069 0032                ;
0070 0032                        ORG     38H     ;INTERRUPT LOCATION
0071 0038 C3 86 01               JMP     TTONE   ;INTERRUPT PROGRAM
0072 003B                ;
0073 003B                ;
0074 003B                ;
0075 003B                MASLP:  EQU     $       ;MASTER LOOP
0076 003B FB                     EI
0077 003C 3A F5 30               LDA     TIME
0078 003F B7                     ORA     A
0079 0040 C2 65 00               JNZ     MAS1    ;TIMING
0080 0043 DB 10                  IN      PORT1
0081 0045 2F                     CMA
0082 0046 E6 20                  ANI     20H     ;150 PTT
0083 0048 CA 58 00               JZ      MAS2
0084 004B 3E C0                  MVI     A,0C0H
0085 004D 32 F6 30       MAS3:   STA     MASK    ;50 MASK
0086 0050 3E 01                  MVI     A,1
0087 0052 32 F5 30               STA     TIME    ;START TIMING
0088 0055 C3 83 00               JMP     ID      ;IDENTIFY
0089 0058 DB 10          MAS2:   IN      PORT1
0090 005A 2F                     CMA
0091 005B E6 10                  ANI     10H     ;450 PTT
0092 005D CA 3B 00               JZ      MASLP   ;NO ACTIVITY
0093 0060 3E 30                  MVI     A,30H   ;450 MASK
0094 0062 C3 4D 00               JMP     MAS3
0095 0065 DB 10          MAS1:   IN      PORT1
0096 0067 2F                     CMA
0097 0068 F5                     PUSH    PSW
0098 0069 E6 20                  ANI     20H     ;150 PTT
0099 006B CA 75 00               JZ      MAS4
0100 006E 21 F6 30               LXI     H,MASK
0101 0071 3E C0                  MVI     A,0C0H
0102 0073 B6                     ORA     M
0103 0074 77                     MOV     M,A
0104 0075 F1             MAS4:   POP     PSW
0105 0076 E6 10                  ANI     10H     ;450 PTT
0106 0078 CA 82 00               JZ      MAS5
0107 007B 21 F6 30               LXI     H,MASK
0108 007E 3E 30                  MVI     A,30H
0109 0080 B6                     ORA     M
0110 0081 77                     MOV     M,A
0111 0082 21 ED 30       MAS5:   LXI     H,TIMER ;INCR 4 BYTE TIMER
0112 0085 34                     INR     M
0113 0086 C2 95 00               JNZ     MAS6
0114 0089 23                     INX     H
0115 008A 34                     INR     M
0116 008B C2 95 00               JNZ     MAS6
0117 008E 23                     INX     H
0118 008F 34                     INR     M
0119 0090 C2 95 00               JNZ     MAS6
0120 0093 23                     INX     H
0121 0094 34                     INR     M
0122 0095 21 ED 30       MAS6:   LXI     H,TIMER ;SEE IF TIME
0123 0098 3E 00                  MVI     A,IDTM0 ;IS UP
0124 009A BE                     CMP     M
0125 009B C2 3B 00               JNZ     MASLP   ;NOPE
0126 009E 23                     INX     H
0127 009F 3E 00                  MVI     A,IDTM1
0128 00A1 BE                     CMP     M
0129 00A2 C2 3B 00               JNZ     MASLP   ;NOPE
0130 00A5 23                     INX     H
0131 00A6 3E 1A                  MVI     A,IDTM2
0132 00A8 BE                     CMP     M
0133 00A9 C2 3B 00               JNZ     MASLP   ;NOPE
0134 00AC 23                     INX     H
0135 00AD 3E 00                  MVI     A,IDTM3
0136 00AF BE                     CMP     M
0137 00B0 C2 3B 00               JNZ     MASLP   ;NOPE
0138 00B3                ;
0139 00B3                ;
0140 00B3                ;
0141 00B3                ID:     EQU     $       ;TIME TO ID IF MASK NONZERO
0142 00B3 21 00 00               LXI     H,0
0143 00B6 22 ED 30               SHLD    TIMER
0144 00B9 22 EF 30               SHLD    TIMER+2
0145 00BC 21 F4 30               LXI     H,IDN
0146 00BF 3A F6 30               LDA     MASK
0147 00C2 B7                     ORA     A
0148 00C3 C2 CC 00               JNZ     ID1
0149 00C6 32 F5 30               STA     TIME    ;STOP TIMING
0150 00C9 C3 3B 00               JMP     MASLP
0151 00CC 3A F3 30       ID1:    LDA     IDS     ;ID STATUS
0152 00CF FE 06                  CPI     6       ;< 6?
0153 00D1 D2 DA 00               JNC     ID2
0154 00D4 32 F4 30               STA     IDN     ;ID NUMBER
0155 00D7 C3 F3 00               JMP     ID3
0156 00DA 34             ID2:    INR     M
0157 00DB 3A F3 30               LDA     IDS
0158 00DE FE 06                  CPI     6
0159 00E0 CA EE 00               JZ      ID4
0160 00E3 3E 05                  MVI     A,5
0161 00E5 BE             ID5:    CMP     M
0162 00E6 D2 F3 00               JNC     ID3
0163 00E9 36 01                  MVI     M,1
0164 00EB C3 F3 00               JMP     ID3
0165 00EE 3E 04          ID4:    MVI     A,4
0166 00F0 C3 E5 00               JMP     ID5
0167 00F3 7E             ID3:    MOV     A,M
0168 00F4 3D                     DCR     A
0169 00F5 07                     RLC
0170 00F6 5F                     MOV     E,A
0171 00F7 16 00                  MVI     D,0
0172 00F9 21 08 10               LXI     H,IDAD
0173 00FC 19                     DAD     D
0174 00FD 5E                     MOV     E,M
0175 00FE 23                     INX     H
0176 00FF 56                     MOV     D,M
0177 0100 EB                     XCHG
0178 0101 CD 08 01               CALL    CW
0179 0104 AF                     XRA     A
0180 0105 32 F6 30               STA     MASK    ;CLEAR MASK
0181 0108 C3 3B 00               JMP     MASLP
0182 0108                ;
0183 0108                ;
0184 0108                ;
0185 0108                ;CW SENDING ROUTINE
0186 0108                ;ROTATES DATA LEFT,
0187 0108                ;1 IS A DASH, 0 IS A DOT
0188 0108                ;UNTIL AN 80 IS LEFT.
0189 0108                ;AN INITIAL 80 IS A SPACE
0190 0108                ;AND A ZERO IS MESSAGE END
0191 0108                ;
0192 0108                ;
0193 0108                CW:     EQU     $       ;CW SENDING ROUTINE
0194 0108 3A F6 30               LDA     MASK
0195 010B E6 50                  ANI     50H
0196 010D 47                     MOV     B,A
0197 010E 11 F9 30               LXI     D,OUTIM
0198 0111 CD 0C 02               CALL    BITS
0199 0114 D3 10                  OUT     PORT1
0200 0116 7E             CW0:    MOV     A,M
0201 0117 B7                     ORA     A
0202 0118 C2 29 01               JNZ     CWND
0203 011B 11 F9 30               LXI     D,OUTIM
0204 011E 06 F0                  MVI     B,0F0H
0205 0120 CD 10 02               CALL    BITC
0206 0123 D3 10                  OUT     PORT1
0207 0125 C9                     RET
0208 0126 FE 80          CWND:   CPI     80H
0209 0128 C2 3B 01               JNZ     CWLET
0210 012B 3E 06                  MVI     A,6     ;SPACE
0211 012D CD 74 01       CW1:    CALL    DELAY
0212 0130 3D                     DCR     A
0213 0131 C2 30 01               JNZ     CW1
0214 0134 23             CW2:    INX     H
0215 0135 C3 16 01               JMP     CW0     ;GET NEXT CHAR
0216 0138 FE 80          CWLET:  CPI     80H     ;DONE?
0217 013A CC 74 01               CZ      DELAY   ;2 SPACES
0218 013D CC 74 01               CZ      DELAY   ;AT END
0219 0140 CA 34 01               JZ      CW2     ;LETTER DONE
0220 0143 F5                     PUSH    PSW
0221 0144 11 F9 30               LXI     D,OUTIM
0222 0147 3A F6 30               LDA     MASK
0223 014A E6 A0                  ANI     0A0H
0224 014C 47                     MOV     B,A
0225 014D CD 0C 02               CALL    BITS
0226 0150 D3 10                  OUT     PORT1   ;OUT TONE
```

Program listing.

It can block access to the control mode, and it can eliminate the ROGER routine. After the LOCK sequence is given, LOAD is called to get three digits. The second digit eliminates the ROGER routine if it is a 1, and the third digit locks the control mode out if it is a 1. ROGER is then called, and LOCK exits. If the second or third digits of a LOCK command are 0, the normal state of the appropriate function is resumed. The LOCK function is intended as a fail-safe measure, available only to the person who constructs the system. The reset instruction (9) is purposely constructed so that it does not reset LCKR, the locker

```
0227 0155 F1                POP     PSW
0228 0156 CD 74 01          CALL    DELAY
0229 0159 17                RAL
0230 015A F5                PUSH    PSW
0231 015B D2 64 01          JNC     CWDOT
0232 015E CD 74 01          CALL    DELAY
0233 0161 CD 74 01          CALL    DELAY
0234 0164 06 A0       CWDOT: MVI    B,0A0H
0235 0166 CD 10 02          CALL    BITC
0236 0169 D3 10             OUT     PORT1
0237 016B CD 74 01          CALL    DELAY
0238 016E F1                POP     PSW     ;GET CHAR
0239 016F E6 FE             ANI     0FEH
0240 0171 C3 3B 01          JMP     CWLET
0241 0174             ;
0242 0174             ;
0243 0174             ;
0244 0174 D5          DELAY: PUSH   D
0245 0175 F5                PUSH    PSW
0246 0176 11 88 13          LXI     D,CWSPD
0247 0179 AF                XRA     A
0248 017A 1B          DEL1:  DCX     D
0249 017B 88                CMP     E
0250 017C C2 7A 01          JNZ     DEL1
0251 017F BA                CMP     D
0252 0180 C2 7A 01          JNZ     DEL1
0253 0183 F1                POP     PSW
0254 0184 D1                POP     D
0255 0185 C9                RET
0256 0186             ;
0257 0186             ;
0258 0186             ;
0259 0186             TTONE: EQU    $
0260 0186             ;INTERRUPT ROUTINE TO SERVICE
0261 0186             ;TOUCH TONE (R) DECODER
0262 0186             ;
0263 0186             ;HANDLES ALL CONTROL AND USER FUNCTIONS
0264 0186             ;
0265 0186             ;
0266 0186 F5                PUSH    PSW
0267 0187 C5                PUSH    B
0268 0188 D5                PUSH    D
0269 0189 E5                PUSH    H
0270 018A 21 00 00          LXI     H,0     ;SEE IF
0271 018D 39                DAD     SP      ;THE STACK
0272 018E 7C                MOV     A,H     ;IS MESSED
0273 018F FE 30             CPI     30H     ;UP
0274 0191 C2 00 00          JNZ     BEGIN   ;YES, RECOVER
0275 0194 3A F6 30          LDA     MASK
0276 0197 F5                PUSH    PSW
0277 0198 11 F9 30          LXI     D,OUTIM
0278 019B 1A                LDAX    D
0279 019C F5                PUSH    PSW
0280 019D 06 F0             MVI     B,0F0H
0281 019F CD 10 02          CALL    BITC    ;BIT CLEAR
0282 01A2 06 01             MVI     B,1
0283 01A4 CD 0C 02          CALL    BITS    ;BIT SET
0284 01A7 D3 10             OUT     PORT1
0285 01A9 DB 20             IN      PORT2
0286 01AB 2F                CMA
0287 01AC E6 02             ANI     2
0288 01AE CA D3 01          JZ      TTON1   ;NOT *
0289 01B1 06 02             MVI     B,2
0290 01B3 CD 0C 02          CALL    BITS
0291 01B6 D3 10             OUT     PORT1   ;KNOCKDOWN
0292 01B8 CD 74 01          CALL    DELAY
0293 01BB CD 10 02          CALL    BITC    ;RELEASE FORCE
0294 01BE D3 10             OUT     PORT1
0295 01C0 11 FB 30          LXI     D,OUT3M
0296 01C3 1A                LDAX    D
0297 01C4 E6 10             ANI     10H     ;LINKED?
0298 01C6 C4 D2 03          CNZ     ROGER   ;YES
0299 01C9 06 10             MVI     B,10H
0300 01CB CD 10 02          CALL    BITC    ;RESTORE TIMER &
0301 01CE D3 30             OUT     PORT3   ;UNLINK RPTRS
0302 01D0 C3 FC 01          JMP     TTON2
0303 01D3 CD 19 02    TTON1: CALL   LOAD
0304 01D6 2A 02 10          LHLD    CDTAB   ;LOCATION OF CODTB
0305 01D9 7E          TTON6: MOV    A,M
0306 01DA B7                ORA     A
0307 01DB CA FC 01          JZ      TTON2   ;NOT CODE
0308 01DE BA                CMP     D
0309 01DF C2 F4 01          JNZ     TTON3
0310 01E2 23                INX     H
0311 01E3 7E                MOV     A,M
0312 01E4 BB                CMP     E
0313 01E5 C2 F5 01          JNZ     TTON4
0314 01E8 23                INX     H
0315 01E9 7E                MOV     A,M
0316 01EA B8                CMP     B
0317 01EB C2 F6 01          JNZ     TTON5
0318 01EE 23                INX     H
0319 01EF 5E                MOV     E,M
0320 01F0 23                INX     H
0321 01F1 56                MOV     D,M
0322 01F2 EB                XCHG            ;ADDR TO HL
0323 01F3 E9                PCHL            ;JUMP TO IT
0324 01F4 23          TTON3: INX    H
0325 01F5 23          TTON4: INX    H
0326 01F6 23          TTON5: INX    H
0327 01F7 23                INX     H
0328 01F8 23                INX     H
0329 01F9 C3 D9 01          JMP     TTON6   ;TRY NEXT CODE
0330 01FC             ;
0331 01FC F1          TTON2: POP    PSW
0332 01FD 32 F9 30          STA     OUTIM
0333 0200 D3 10             OUT     PORT1
0334 0202 F1                POP     PSW
0335 0203 32 F6 30          STA     MASK
0336 0206 E1                POP     H
0337 0207 D1                POP     D
0338 0208 C1                POP     B
0339 0209 F1                POP     PSW
0340 020A FB                EI
0341 020B C9                RET             ;AND EXIT INTERRUPT
0342 020C             ;
0343 020C             ;
0344 020C             ;
0345 020C             ;
0346 020C             ;
0347 020C             ;
0348 020C             BITS:  EQU    $
0349 020C             ;BIT SET SETS THE BIT CORRESPONDING
0350 020C             ;TO THAT OF REG B IN MEM LOC (DE)
0351 020C             ;
0352 020C 1A                LDAX    D
0353 020D B0                ORA     B
0354 020E 12                STAX    D
0355 020F C9                RET
0356 0210             ;
0357 0210             ;
0358 0210             BITC:  EQU    $
0359 0210             ;BIT CLEAR CLEARS THE BIT
0360 0210 C5                PUSH    B
0361 0211 78                MOV     A,B
0362 0212 2F                CMA
0363 0213 47                MOV     B,A
0364 0214 1A                LDAX    D
0365 0215 A0                ANA     B
0366 0216 12                STAX    D
0367 0217 C1                POP     B
0368 0218 C9                RET
0369 0219             ;
0370 0219             ;
0371 0219             ;
0372 0219             LOAD:  EQU    $
0373 0219             ;LOAD PUTS THE SEQUENTIAL 3 DIGIT CODE
0374 0219             ;INTO REG'S D,E,B
0375 0219             ;REG'S CLEAR IF NO DIGIT
0376 0219             ;LOAD WAITS FOR A LONG FIRST TONE
0377 0219             ;AND THEN FAST TWO TONES
0378 0219             ;
0379 0219 11 00 00          LXI     D,0
0380 021C 01 00 00          LXI     B,0
0381 021F DB 10       LOAD1: IN     PORT1
0382 0221 2F                CMA
0383 0222 B7                ORA     A
0384 0223 F2 1F 02          JP      LOAD1   ;WAIT FOR LONG TONE
0385 0226 CD 3B 02          CALL    DECOD   ;GET DIGIT
0386 0229 57                MOV     D,A
0387 022A CD 68 02          CALL    HURRY
0388 022D D8                RC              ;TIMED OUT
0389 022E CD 3B 02          CALL    DECOD
0390 0231 5F                MOV     E,A
0391 0232 CD 68 02          CALL    HURRY
0392 0235 D8                RC
0393 0236 CD 3B 02          CALL    DECOD
0394 0239 47                MOV     B,A
0395 023A C9                RET
0396 023B             ;
0397 023B             ;
0398 023B             ;
0399 023B             DECOD: EQU    $
0400 023B             ;DECOD READS THE TOUCH TONE (R) DECODER
0401 023B             ;AND PUTS THE CHAR IN REG A
0402 023B             ;NO TONE RETURNS A ZERO
0403 023B             ;WAIT FOR RELEASE OF TONE BEFORE RETURN
0404 023B             ;
0405 023B C5                PUSH    B
0406 023C DB 10             IN      PORT1
0407 023E 2F                CMA
0408 023F E6 0F             ANI     0FH
0409 0241 C2 50 02          JNZ     DECD1   ;IN PORT 1
0410 0244 DB 20             IN      PORT2
0411 0246 2F                CMA
0412 0247 B7                ORA     A
0413 0248 CA 5C 02          JZ      DECD4   ;ERRONEOUS CHDX
0414 024B 0E 04             MVI     C,4     ;OFFSET
0415 024D C3 56 02          JMP     DECD2
0416 0250 17          DECD1: RAL
0417 0251 17                RAL
0418 0252 17                RAL
0419 0253 17                RAL
0420 0254 0E 00             MVI     C,0
0421 0256 0C          DECD2: INR    C
0422 0257 17                RAL
0423 0258 D2 56 02          JNC     DECD2
0424 025B 79                MOV     A,C
0425 025C C1          DECD4: POP    B
0426 025D F5                PUSH    PSW
0427 025E DB 10       DECD3: IN     PORT1
0428 0260 2F                CMA
0429 0261 E6 A0             ANI     40H
0430 0263 C2 5E 02          JNZ     DECD3
0431 0266 F1                POP     PSW
0432 0267 C9                RET
0433 0268             ;
0434 0268             ;
0435 0268             ;
0436 0268             HURRY: EQU    $
0437 0268             ;HURRY WAITS FOR A TONE,
0438 0268             ;BUT EXITS WITH CARRY SET IF NONE
0439 0268             ;RECEIVED IN PERMITTED TIME
0440 0268             ;
0441 0268 AF                XRA     A
0442 0269 D5                PUSH    D
0443 026A C5                PUSH    B
0444 026B 11 00 00          LXI     D,0
0445 026E 01 00 00          LXI     B,0
0446 0271 DB 10       HURY1: IN     PORT1
0447 0273 2F                CMA
0448 0274 E6 40             ANI     40H     ;TONE?
```

Continued on next page

where the control mode may be inhibited.

PATCH, the autopatch routine, is one of the more complicated subprograms. Shown in Fig. 11, PATCH first checks to see if the autopatch is enabled. NOTIM (no timer) is cleared so that the timer will be present unless changed later. GETTT gets the requested telephone number. The digit count is then checked. If no number was sent, and a direct autopatch is allowed, then at PTCH1 AP is pulsed, giving the user the line to dial his own number. Otherwise, the attempt is aborted. If 7 digits were entered, control passes through PTCH2 to PTCH8. If the first digit of the number is a 1, the patch is aborted. If not, at PTCH3 AP is pulsed, bringing up the line. At PTCH5, a one-second delay is introduced to allow time for the telephone company equipment to produce the dial tone.

```
0449 0276 C2 8D 02          JNZ     HURY3   ;YES
0450 0279 0C                INR     C
0451 027A C2 86 02          JNZ     HURY2
0452 027D 04                INR     B
0453 027E C2 86 02          JNZ     HURY2
0454 0281 1C                INR     E
0455 0282 C2 86 02          JNZ     HURY2
0456 0285 14                INR     D
0457 0286 3E 01     HURY2:  MVI     A,1
0458 0288 B8                CMP     E
0459 0289 C2 71 02          JNZ     HURY1
0460 028C 37                STC             ;TIMED OUT
0461 028D C1        HURY3:  POP     B
0462 028E D1                POP     D
0463 028F C9                RET
0464 0290           ;
0465 0290           ;
0466 0290           ;
0467 0290           WCD:    EQU     $
0468 0290           ;WAIT FOR CARRIER DROP
0469 0290 F5                PUSH    PSW
0470 0291 3A F7 30          LDA     LKROG
0471 0294 B7                ORA     A
0472 0295 C2 B5 02          JNZ     WCD2
0473 0298 DB 30             IN      PORT3
0474 029A B7                ORA     A
0475 029B FA B5 02          JM      WCD2
0476 029E E6 40             ANI     40H
0477 02A0 CA AD 02          JZ      WCD1    ;NOT CR
0478 02A3 DB 30     WCD3:   IN      PORT3
0479 02A5 E6 40             ANI     40H
0480 02A7 C2 A3 02          JNZ     WCD3
0481 02AA C3 B5 02          JMP     WCD2
0482 02AD DB 10     WCD1:   IN      PORT1
0483 02AF 2F                CMA
0484 02B0 E6 20             ANI     20H
0485 02B2 C2 AD 02          JNZ     WCD1
0486 02B5 F1        WCD2:   POP     PSW
0487 02B6 C9                RET
0488 02B7           ;
0489 02B7           ;
0490 02B7           ;
0491 02B7 3A F8 30  LINK:   LDA     OUTOM   ;LINK RPTRS
0492 02BA E6 01             ANI     1       ;ENABLED?
0493 02BC C2 FC 01          JNZ     TTOH2   ;NO!
0494 02BF 11 F8 30          LXI     D,OUT3M
0495 02C2 06 10             MVI     B,10H
0496 02C4 CD 0C 02          CALL    BITS
0497 02C7 D3 30             OUT     PORT3
0498 02C9 CD D2 03          CALL    ROGER
0499 02CC C3 FC 01          JMP     TTOH2
0500 02CF           ;
0501 02CF           ;
0502 02CF           ;
0503 02CF 3A F8 30  TAPE:   LDA     OUTOM   ;TAPE LOOP
0504 02D2 E6 02             ANI     2       ;ENABLED?
0505 02D4 C2 FC 01          JNZ     TTOH2   ;NO!
0506 02D7 CD 90 02  TAP1:   CALL    WCD
0507 02DA 11 F8 30          LXI     D,OUT3M
0508 02DD 06 80             MVI     B,80H
0509 02DF CD 0C 02          CALL    BITS
0510 02E2 D3 30             OUT     PORT3
0511 02E4 CD 74 01          CALL    DELAY
0512 02E7 CD 10 02          CALL    BITC    ;PULSE TAPE
0513 02EA D3 30             OUT     PORT3
0514 02EC C3 FC 01          JMP     TTOH2
0515 02EF           ;
0516 02EF           ;
0517 02EF           ;
0518 02EF           ;SELECTIVE CALL DISABLES TONE BLOCKING
0519 02EF 11 F9 30  SELCL:  LXI     D,OUT1M
0520 02F2 06 01             MVI     B,1
0521 02F4 CD 10 02          CALL    BITC
0522 02F7 D3 10             OUT     PORT1
0523 02F9 CD 90 02          CALL    WCD     ;WAIT FOR CAR DROP
0524 02FC C3 FC 01          JMP     TTOH2
0525 02FF           ;
0526 02FF           ;
0527 02FF           ;
0528 02FF           CNTRL:  EQU     $
0529 02FF           ;ALL CONTROL CODES ARE ENTERED
0530 02FF           ;IN THIS ROUTINE.
0531 02FF           ;
0532 02FF 3A F2 30          LDA     LCKR
0533 0302 B7                ORA     A
0534 0303 C2 FC 01          JNZ     TTOH2   ;LOCKED OUT
0535 0306 CD 90 02          CALL    WCD
0536 0309 CD 19 02          CALL    LOAD
0537 030C 2A 04 10          LHLD    CFRMC
0538 030F C3 D9 01          JMP     TTOH6
0539 0312 CD 90 02  CNTR0:  CALL    WCD
0540 0315 DB 10     CNTR1:  IN      PORT1
0541 0317 2F                CMA
0542 0318 B7                ORA     A
0543 0319 F2 15 03          JP      CNTR1
0544 031C CD 38 02          CALL    DECOD   ;GET DIGIT
0545 031F CD 90 02          CALL    WCD
0546 0322 B7                ORA     A
0547 0323 CA FC 01          JZ      TTOH2
0548 0326 FE 0D             CPI     13
0549 0328 D2 FC 01          JNC     TTOH2   ;INVALID
0550 032B CD D2 03          CALL    ROGER
0551 032E FE 08             CPI     8
0552 0330 D2 39 03          JNC     CNTR2
0553 0333 32 F3 30          STA     IDS
0554 0336 C3 FC 01          JMP     TTOH2   ;DONE
0555 0339 FE 08     CNTR2:  CPI     8
0556 033B CA A0 10          JZ      IDLD    ;LOAD ID
0557 033E FE 0A             CPI     10
0558 0340 CA 50 03          JZ      OUT
0559 0343 FE 09             CPI     9
0560 0345 CA 04 00          JZ      RESET   ;INIT
0561 0348 FE 0B             CPI     11
0562 034A CA F8 03          JZ      IDENT
0563 034D C3 12 10          JMP     LNUMA   ;LOAD NUMBER
0564 0350           ;
0565 0350           ;
0566 0350           OUT:    EQU     $
0567 0350           ;OUT TAKES 3 DIGITS AND OUTPUTS
0568 0350           ;TO OUTPUT PORTS AS FOLLOWS:
0569 0350           ;TT CODE IS XYZ
0570 0350           ;PORT X         0,1,2,3,4,5,6,7
0571 0350           ;BIT Y          1,2,3,4,5,6,7,8
0572 0350           ;LEVEL Z        0,1
0573 0350           ;
0574 0350           ;ALL FAULTY CODES ARE IGNORED
0575 0350           ;
0576 0350           ;A * EXITS THE CONTROL-OUT MODE
0577 0350           ;
0578 0350 CD 19 02          CALL    LOAD
0579 0353 3E 0B             MVI     A,11    ;*
0580 0355 BA                CMP     D
0581 0356 C2 5F 03          JNZ     OUT1
0582 0359 CD D2 03          CALL    ROGER
0583 035C C3 FC 01          JMP     TTOH2   ;EXIT
0584 035F AF        OUT1:   XRA     A
0585 0360 BA                CMP     D
0586 0361 CA CC 03          JZ      OUTR    ;NO GOOD
0587 0364 BB                CMP     E
0588 0365 CA CC 03          JZ      OUTR
0589 0368 7A                MOV     A,D
0590 0369 FE 0A             CPI     10
0591 036B C2 73 03          JNZ     OUT4
0592 036E 16 00             MVI     D,0
0593 0370 C3 79 03          JMP     OUT5
0594 0373 3E 07     OUT4:   MVI     A,7
0595 0375 BA                CMP     D
0596 0376 DA CC 03          JC      OUTR
0597 0379 3E 08     OUT5:   MVI     A,8
0598 037B BB                CMP     E
0599 037C DA CC 03          JC      OUTR
0600 037F 3E 01             MVI     A,1
0601 0381 B8                CMP     B
0602 0382 CA 8E 03          JZ      OUT3
0603 0385 3E 0A             MVI     A,10
0604 0387 B8                CMP     B
0605 0388 CA 8E 03          JZ      OUT3
0606 038B C3 CC 03          JMP     OUTR
0607 038E CD D2 03  OUT3:   CALL    ROGER
0608 0391 3E 80             MVI     A,80H
0609 0393 07        OUT2:   RLC
0610 0394 1D                DCR     E
0611 0395 C2 93 03          JNZ     OUT2
0612 0398 5F                MOV     E,A
0613 0399 7A        OPRT:   MOV     A,D     ;PORT #
0614 039A 07                RLC             ;TIMES 10H
0615 039B 07                RLC
0616 039C 07                RLC
0617 039D 07                RLC
0618 039E 32 EB 30          STA     OUTR2
0619 03A1 78                MOV     A,B     ;LEVEL
0620 03A2 43                MOV     B,E     ;BIT #
0621 03A3 3D                DCR     A
0622 03A4 F5                PUSH    PSW
0623 03A5 7A                MOV     A,D
0624 03A6 11 F8 30          LXI     D,OUTOM
0625 03A9 83                ADD     E
0626 03AA 5F                MOV     E,A
0627 03AB 7A                MOV     A,D
0628 03AC CE 00             ACI     0
0629 03AE 57                MOV     D,A
0630 03AF 3E D3             MVI     A,0D3H
0631 03B1 32 EA 30          STA     OUTR1   ;OUT INSTR
0632 03B4 3E C9             MVI     A,0C9H
0633 03B6 32 EC 30          STA     OUTR3   ;RETURN INSTR
0634 03B9 F1                POP     PSW
0635 03BA CA C6 03          JZ      OPRT1
0636 03BD CD 10 02  OPRT0:  CALL    BITC
0637 03C0 CD EA 30  OPRT2:  CALL    OUTR1   ;OUTPUT
0638 03C3 C7 50 03          JMP     OUT     ;GET NEXT CMD
0639 03C6 CD 0C 02  OPRT1:  CALL    BITS
0640 03C9 C3 C0 03          JMP     OPRT2
0641 03CC CD 90 02  OUTR:   CALL    WCD
0642 03CF C3 50 03          JMP     OUT
0643 03D2           ;
0644 03D2           ;
0645 03D2           ;
0646 03D2           ROGER:  EQU     $
0647 03D2           ;ROGER SENDS AN 'R' IN MORSE
0648 03D2 CD 90 02          CALL    WCD
0649 03D5 F5                PUSH    PSW
0650 03D6 3A F7 30          LDA     LKROG
0651 03D9 B7                ORA     A
0652 03DA C2 F6 03          JNZ     ROG1
0653 03DD E5                PUSH    H
0654 03DE D5                PUSH    D
0655 03DF C5                PUSH    B
0656 03E0 3A F6 30          LDA     MASK
0657 03E3 F5                PUSH    PSW
0658 03E4 3E C0             MVI     A,0C0H
0659 03E6 32 F6 30          STA     MASK
0660 03E9 2A 06 10          LHLD    RMSGA   ;LOC OF RMSG
0661 03EC CD 0B 01          CALL    CW
0662 03EF F1                POP     PSW
0663 03F0 32 F6 30          STA     MASK
0664 03F3 C1                POP     B
0665 03F4 D1                POP     D
0666 03F5 E1                POP     H
0667 03F6 F1        ROG1:   POP     PSW
0668 03F7 C9                RET
0669 03F8           ;
0670 03F8           ;
0671 03F8           ;
0672 03F8           ;FORCE A REPEATER IDENTIFICATION
0673 03F8 AF        IDENT:  XRA     A
0674 03F9 32 F5 30          STA     TIME    ;FORCE AN ID
```

Our exchange is an electronic switching system and is very rapid. If it commonly takes longer than one second at your exchange, change the number 15 to a larger number in line #1057. A 1 is sent to the LD output, preparing to dial the number. At PTCH6, the number is dialed. Each tone is on for 65 ms and off for 65 ms, the time DELAY waits. The binary digit numbers are converted to the proper row and column format by the TTTAB (touchtone table). When the number is completed, LD is turned off, and if NOTIM is not 0, the timer is disabled. Similarly, if 8 or 11 digits are requested and the first digit is a 0, the same procedure applies. If a single-digit number is requested, a table is searched at PCH10. The single digit table, SDTAB, has the single digit followed by the address of the corresponding telephone number. At the loca-

```
0675 03FC C3 FC 01          JMP     TTON2
0676 03FF                   ;
0677 03FF                   ;
0678 03FF                   ;
0679 03FF                   ORG     1000H   ;SECOND ROM
0680 1000 EA 30         STCK:   DW      STACK
0681 1002 15 10         CDTAB:  DW      CODTB
0682 1004 49 10         CFMCD:  DW      CFMCD
0683 1006 9A 10         RMSGA:  DW      RMSG
0684 1008 4E 10         IDAD:   DW      IDAD1
0685 100A 5F 10                 DW      IDAD2
0686 100C 75 10                 DW      IDAD3
0687 100E 84 10                 DW      IDAD4
0688 1010 25 30                 DW      IDAD5
0689 1012 C3 12 13      LNUMA:  JMP     LNUM
0690 1015                   ;
0691 1015 09            CODTB:  DB      6
0692 1016 08                    DB      7
0693 1017 0C                    DB      12      ;#
0694 1018 83 11                 DW      PATCH
0695 101A 03                    DB      9
0696 101B 0C                    DB      12      ;#
0697 101C 08                    DB      5
0698 101D AF 12                 DW      RBASE
0699 101F 03                    DB      3
0700 1020 0C                    DB      12      ;#
0701 1021 03                    DB      3
0702 1022 EF 02                 DW      SELCL
0703 1024 01                    DB      1
0704 1025 0C                    DB      12      ;#
0705 1026 01                    DB      1
0706 1027 B7 02                 DW      LINK
0707 1029 02                    DB      2
0708 102A 0C                    DB      12      ;#
0709 102B 02                    DB      2
0710 102C CF 02                 DW      TAPE
0711 102E 04                    DB      4
0712 102F 0C                    DB      12      ;#
0713 1030 04                    DB      4
0714 1031 F3 10                 DW      TTTST
0715 1033 05                    DB      5
0716 1034 0C                    DB      12      ;#
0717 1035 05                    DB      5
0718 1036 C1 12                 DW      DIAL
0719 1038 0C                    DB      6
0720 1039 02                    DB      11      ;*
0721 103A 0B                    DB      12      ;#
0722 103B FF 02                 DW      .CNTRL
0723 103D 0C                    DB      12      ;#
0724 103E 05                    DB      4
0725 103F 0B                    DB      8
0726 1040 D7 12                 DW      LOCK
0727 1042 0C                    DB      2
0728 1043 0B                    DB      11      ;*
0729 1044 02                    DB      2
0730 1045 B7 12                 DW      TAP2
0731 1047 00                    DB      0
0732 1048 0B            CFMCD:  DB      12      ;#
0733 1049 02                    DB      11      ;*
0734 104A 0C                    DB      6
0735 104B 12 03                 DW      CNTRO
0736 104D 00                    DB      0
0737 104E                   ;
0738 104E                   ;
0739 104E                   ;
0740 104E 80            IDAD1:  DB      80H
0741 104F 90                    DB      90H     ;D
0742 1050 40                    DB      40H     ;E
0743 1051 80                    DB      80H     ;SP
0744 1052 70                    DB      70H     ;W
0745 1053 50                    DB      50H     ;R
0746 1054 1C                    DB      1CH     ;3
0747 1055 60                    DB      60H     ;A
0748 1056 28                    DB      28H     ;F
0749 1057 E0                    DB      0E0H    ;M
0750 1058 80                    DB      80H     ;SP
0751 1059 88                    DB      88H     ;B
0752 105A 60                    DB      60H     ;A
0753 105B 50                    DB      50H     ;R
0754 105C A8                    DB      0A8H    ;C
0755 105D 80                    DB      80H     ;SP
0756 105E 00                    DB      0
0757 105F 80            IDAD2:  DB      80H     ;SP
0758 1060 90                    DB      90H     ;D
0759 1061 40                    DB      40H     ;E
0760 1062 80                    DB      80H     ;SP
0761 1063 70                    DB      70H     ;W
0762 1064 50                    DB      50H     ;R
0763 1065 1C                    DB      1CH     ;3
0764 1066 60                    DB      60H     ;A
0765 1067 28                    DB      28H     ;F
0766 1068 E0                    DB      0E0H    ;M
0767 1069 80                    DB      80H     ;SP
0768 106A 88                    DB      88H     ;B
0769 106B 60                    DB      60H     ;A
0770 106C 48                    DB      48H     ;L
0771 106D C0                    DB      0C0H    ;T
0772 106E 20                    DB      20H     ;I
0773 106F E0                    DB      0E0H    ;M
0774 1070 F0                    DB      0F0H    ;O
0775 1071 50                    DB      50H     ;R
0776 1072 40                    DB      40H     ;E
0777 1073 80                    DB      80H     ;SP
0778 1074 00                    DB      0
0779 1075 80            IDAD3:  DB      80H     ;SP
0780 1076 C4                    DB      0C4H    ;7
0781 1077 1C                    DB      1CH     ;3
0782 1078 80                    DB      80H     ;SP
0783 1079 90                    DB      90H     ;D
0784 107A 40                    DB      40H     ;E
0785 107B 80                    DB      80H     ;SP
0786 107C 70                    DB      70H     ;W
0787 107D 50                    DB      50H     ;R
0788 107E 1C                    DB      1CH     ;3
0789 107F 60                    DB      60H     ;A
0790 1080 28                    DB      28H     ;F
0791 1081 E0                    DB      0E0H    ;M
0792 1082 80                    DB      80H     ;SP
0793 1083 00                    DB      0
0794 1084 80            IDAD4:  DB      80H     ;SP
0795 1085 90                    DB      90H     ;D
0796 1086 40                    DB      40H     ;E
0797 1087 80                    DB      80H     ;SP
0798 1088 70                    DB      70H     ;W
0799 1089 50                    DB      50H     ;R
0800 108A 1C                    DB      1CH     ;3
0801 108B 60                    DB      60H     ;A
0802 108C 28                    DB      28H     ;F
0803 108D E0                    DB      0E0H    ;M
0804 108E 80                    DB      80H     ;SP
0805 108F 88                    DB      88H     ;B
0806 1090 60                    DB      60H     ;A
0807 1091 48                    DB      48H     ;L
0808 1092 C0                    DB      0C0H    ;T
0809 1093 F0                    DB      0F0H    ;O
0810 1094 80                    DB      80H     ;SP
0811 1095 60                    DB      60H     ;A
0812 1096 50                    DB      50H     ;R
0813 1097 A8                    DB      0A8H    ;C
0814 1098 80                    DB      80H     ;SP
0815 1099 00                    DB      0
0816 109A 80            RMSG:   DB      80H     ;SP
0817 109B 80                    DB      80H     ;SP
0818 109C 80                    DB      80H     ;SP
0819 109D 50                    DB      50H     ;R
0820 109E 80                    DB      80H     ;SP
0821 109F 00                    DB      0
0822 10A0                   ;
0823 10A0                   ;
0824 10A0                   ;
0825 10A0               IDLD:   EQU     $
0826 10A0               ;ID LOAD LOADS A CW ID INTO RAM
0827 10A0               ;THIS CORRESPONDS TO ID #5
0828 10A0               ;
0829 10A0               ;0 IS DIT, 1 IS DAH
0830 10A0               ;2 IS END CHARACTER
0831 10A0               ;3 IS END ID
0832 10A0               ;
0833 10A0 21 25 30          LXI     H,IDAD5
0834 10A3 06 00         IDLDO:  MVI     B,0
0835 10A5 0E 00             MVI     C,0     ;ELEMENT COUNT
0836 10A7 DB 10         IDLD1:  IN      PORT1
0837 10A9 2F                CMA
0838 10AA E6 40             ANI     40H
0839 10AC CA A7 10          JZ      IDLD1
0840 10AF CD 3B 02          CALL    DECOD
0841 10B2 FE 03             CPI     3
0842 10B4 C2 BF 10          JNZ     IDNTS
0843 10B7 36 00             MVI     M,0
0844 10B9 CD D2 03          CALL    ROGER
0845 10BC C3 FC 01          JMP     TTON2
0846 10BF FE 02         IDNTS:  CPI     2
0847 10C1 C2 D9 10          JNZ     IDCH
0848 10C4 78                MOV     A,B
0849 10C5 37                STC             ;LEFT JUSTIFY
0850 10C6 17            IDDLT:  RAL
0851 10C7 47                MOV     B,A
0852 10C8 0C                INR     C
0853 10C9 3E 07             MVI     A,7
0854 10CB B9                CMP     C
0855 10CC DA D4 10          JC      IDDL
0856 10CF AF                XRA     A
0857 10D0 78                MOV     A,B
0858 10D1 C3 C6 10          JMP     IDDLT
0859 10D4 70            IDDL:   MOV     M,B
0860 10D5 23                INX     H
0861 10D6 C3 A3 10          JMP     IDLDO
0862 10D9 FE 01         IDCH:   CPI     1
0863 10DB C2 E6 10          JNZ     IDCHD
0864 10DE 78                MOV     A,B
0865 10DF 37                STC
0866 10E0 17                RAL
0867 10E1 47                MOV     B,A
0868 10E2 0C                INR     C
0869 10E3 C3 A7 10          JMP     IDLD1
0870 10E6 FE 0A         IDCHD:  CPI     10
0871 10E8 C2 A7 10          JNZ     IDLD1   ;INVALID
0872 10EB AF                XRA     A
0873 10EC 78                MOV     A,B
0874 10ED 17                RAL
0875 10EE 47                MOV     B,A
0876 10EF 0C                INR     C
0877 10F0 C3 A7 10          JMP     IDLD1
0878 10F3                   ;
0879 10F3                   ;
0880 10F3                   ;
0881 10F3               TTTST:  EQU     $
0882 10F3               ;TOUCH TONE (R) TEST ROUTINE
0883 10F3               ;THIS FUNCTION PLACES UP TO 24 DIGITS
0884 10F3               ;INTO A BUFFER AND REPEATS WHAT IT RECEIVED
0885 10F3               ;IN CW AFTER CARRIER DROP
0886 10F3               ;
0887 10F3 3A F8 30          LDA     OUTON   ;ENABLED?
0888 10F6 E6 08             ANI     8
0889 10F8 C2 FC 01          JNZ     TTON2   ;NO!
0890 10FB CD 89 11          CALL    GETTT   ;GET TOUCH TONE (R)
0891 10FE 11 01 30      TTST0:  LXI     D,TTDIG+1
0892 1101 3A 00 30          LDA     TTDIG
0893 1104 F5                PUSH    PSW     ;SAVE IT
0894 1105 3A 00 30      TTST1:  LDA     TTDIG
0895 1108 B7                ORA     A
0896 1109 C2 13 11          JNZ     TTST2
```

Continued on next page

tion of the number, the number of digits precedes the actual number, permitting any digit length. A 0 must be stored as a decimal 10. If the number is not found, PATCH exits. If found, the digit count is checked, primarily for the programmable number. If the number is valid, the telephone number is copied into the GETTT buffer, NOTIM is set, and control goes to PTCH3, where the rest is normal.

The remote base routine, RBASE, merely pulses RB. TAP2, the secondary tape access, jumps to the appropriate point in TAPE. DIAL, the 5#5 function, makes various checks and jumps to TTTST at a point where the existing buffer is sent.

The two ROMs are set up in a fashion to permit as many changes as possible in the second ROM without requiring a replacement of the first ROM as well. Most

```
0897 110C F1            TTST2:  POP     PSW
0898 110D 32 00 30              STA     TTDIG   ;RESTORE IT
0899 1110 C3 FC 01              JMP     TTON2   ;DONE
0900 1113 3D            TTST1:  DCR     A
0901 1114 32 00 30              STA     TTDIG
0902 1117 D5                    PUSH    D
0903 1118 21 87 11              LXI     H,CWDSP ;SPACE
0904 111B 3A F6 30              LDA     MASK
0905 111E F5                    PUSH    PSW
0906 111F 3E C0                 MVI     A,0C0H  ;150 ONLY
0907 1121 32 F6 30              STA     MASK
0908 1124 CD 0B 01              CALL    CW
0909 1127 F1                    POP     PSW
0910 1128 32 F6 30              STA     MASK
0911 112B D1                    POP     D
0912 112C D5                    PUSH    D
0913 112D 1A                    LDAX    D       ;GET DIGIT
0914 112E 3D                    DCR     A
0915 112F 07                    RLC             ;TIMES 2
0916 1130 5F                    MOV     E,A
0917 1131 16 00                 MVI     D,0
0918 1133 21 50 11              LXI     H,DIGAD
0919 1136 19                    DAD     D
0920 1137 5E                    MOV     E,M
0921 1138 23                    INX     H
0922 1139 56                    MOV     D,M
0923 113A EB                    XCHG            ;ADDR TO HL
0924 113B 3A F6 30              LDA     MASK
0925 113E F5                    PUSH    PSW
0926 113F 3E C0                 MVI     A,0C0H
0927 1141 32 F6 30              STA     MASK
0928 1144 CD 0B 01              CALL    CW
0929 1147 F1                    POP     PSW
0930 1148 32 F6 30              STA     MASK
0931 114B D1                    POP     D
0932 114C 13                    INX     D
0933 114D C3 05 11              JMP     TTST1
0934 1150               ;
0935 1150               ;
0936 1150               ;
0937 1150 69 11         DIGAD:  DW      CWD1
0938 1152 6A 11                 DW      CWD2
0939 1154 6C 11                 DW      CWD3
0940 1156 6E 11                 DW      CWD4
0941 1158 70 11                 DW      CWD5
0942 115A 72 11                 DW      CWD6
0943 115C 74 11                 DW      CWD7
0944 115E 76 11                 DW      CWD8
0945 1160 78 11                 DW      CWD9
0946 1162 7A 11                 DW      CWD0
0947 1164 7C 11                 DW      CWDS
0948 1166 81 11                 DW      CWDP
0949 1168               ;
0950 1168               ;
0951 1168               ;
0952 1168 7C            CWD1:   DB      7CH
0953 1169 00                    DB      0
0954 116A 3C            CWD2:   DB      3CH
0955 116B 00                    DB      0
0956 116C 1C            CWD3:   DB      1CH
0957 116D 00                    DB      0
0958 116E 0C            CWD4:   DB      0CH
0959 116F 00                    DB      0
0960 1170 04            CWD5:   DB      4
0961 1171 00                    DB      0
0962 1172 84            CWD6:   DB      84H
0963 1173 00                    DB      0
0964 1174 C4            CWD7:   DB      0C4H
0965 1175 00                    DB      0
0966 1176 E4            CWD8:   DB      0E4H
0967 1177 00                    DB      0
0968 1178 F4            CWD9:   DB      0F4H
0969 1179 00                    DB      0
0970 117A FC            CWD0:   DB      0FCH
0971 117B 00                    DB      0
0972 117C 10            CWDS:   DB      10H     ;S
0973 117D C0                    DB      0C0H    ;T
0974 117E 60                    DB      60H     ;A
0975 117F 50                    DB      50H     ;R
0976 1180 00                    DB      0
0977 1181 68            CWDP:   DB      68H     ;P
0978 1182 F0                    DB      0F0H    ;O
0979 1183 30                    DB      30H     ;U
0980 1184 A0                    DB      0A0H    ;N
0981 1185 90                    DB      90H     ;D
0982 1186 00                    DB      0
0983 1187 80            CWDSP:  DB      80H
0984 1188 00                    DB      0
0985 1189               ;
0986 1189               ;
0987 1189               ;
0988 1189               GETTT:  EQU     $
0989 1189               ;GET TOUCH TONE (R) ROUTINE
0990 1189               ;PLACES UP TO 24 DIGITS IN BUFFER
0991 1189               ;AT TTDIG+1, DIGIT COUNT AT TTDIG
0992 1189 11 00 30              LXI     D,TTDIG
0993 118C AF                    XRA     A
0994 118D 12                    STAX    D
0995 118E 13                    INX     D
0996 118F DB 10         GETT1:  IN      PORT1
0997 1191 E6 20                 ANI     20H
0998 1193 C0                    RNZ             ;CARRIER GONE
0999 1194 DB 10                 IN      PORT1
1000 1196 E6 40                 ANI     40H
1001 1198 C2 8F 11              JNZ     GETT1   ;NO TONE
1002 119B CD 3B 02              CALL    DECOD
1003 119E B7                    ORA     A
1004 119F CA 8F 11              JZ      GETT1
1005 11A2 12                    STAX    D       ;STORE DIGIT
1006 11A3 13                    INX     D
1007 11A4 3A 00 30              LDA     TTDIG
1008 11A7 3C                    INR     A
1009 11A8 32 0030               STA     TTDIG
1010 11AB FE 18                 CPI     24
1011 11AD C2 8F 11              JNZ     GETT1
1012 11B0 C3 90 02              JMP     WCD
1013 11B3               ;
1014 11B3               ;
1015 11B3               ;
1016 11B3               ;AUTOPATCH ROUTINE - CHECKS REQUESTED
1017 11B3               ;NUMBER FOR VALIDITY, IF OK IT
1018 11B3               ;BRINGS UP LINE AND REDIALS THE NUMBER
1019 11B3 3A FA 30      PATCH:  LDA     OUT2M
1020 11B6 E6 20                 ANI     20H
1021 11B8 C2 FC 01              JNZ     TTON2
1022 11BB AF                    XRA     A
1023 11BC 32 F1 30              STA     NOTIM
1024 11BF CD 89 11              CALL    GETTT   ;GET NUMBER
1025 11C2 3A 00 30              LDA     TTDIG
1026 11C5 B7                    ORA     A
1027 11C6 C2 E6 11              JNZ     PTCH2
1028 11C9 3A F8 30              LDA     OUTOM   ;DIRECT
1029 11CC E6 04                 ANI     4       ;ENABLED?
1030 11CE CA FC 01              JZ      TTON2   ;NO!
1031 11D1 11 F9 30              LXI     D,OUT1M
1032 11D4 06 08                 MVI     B,8
1033 11D6 CD 0C 02      PTCH1:  CALL    BITS
1034 11D9 D3 10                 OUT     PORT1
1035 11DB CD 74 01              CALL    DELAY
1036 11DE CD 10 02              CALL    BITC
1037 11E1 D3 10                 OUT     PORT1
1038 11E3 C3 FC 01              JMP     TTON2
1039 11E6 FE 01         PTCH2:  CPI     1
1040 11E8 CA 75 12              JZ      PTCH9
1041 11EB FE 07                 CPI     7
1042 11ED CA 6A 12              JZ      PTCH8
1043 11F0 FE 0B                 CPI     11
1044 11F2 CA FA 11              JZ      PTCH4
1045 11F5 FE 08                 CPI     8
1046 11F7 C2 FC 01              JNZ     TTON2
1047 11FA 3A 01 30      PTCH4:  LDA     TTDIG+1
1048 11FD FE 0A                 CPI     10      ;ZERO
1049 11FF C2 FC 01              JNZ     TTON2   ;NOT COLLECT
1050 1202 11 F9 30      PTCH3:  LXI     D,OUT1M
1051 1205 06 08                 MVI     B,8
1052 1207 CD 0C 02              CALL    BITS
1053 120A D3 10                 OUT     PORT1
1054 120C CD 74 01              CALL    DELAY
1055 120F CD 10 02              CALL    BITC
1056 1212 D3 10                 OUT     PORT1
1057 1214 3E 0F                 MVI     A,15
1058 1216 CD 74 01      PTCH5:  CALL    DELAY
1059 1219 3D                    DCR     A
1060 121A C2 16 12              JNZ     PTCH5   ;1 SEC WAIT
1061 121D 11 FB 30              LXI     D,OUT3M
1062 1220 06 40                 MVI     B,40H
1063 1222 CD 0C 02              CALL    BITS    ;TURN LINE ON
1064 1225 D3 30                 OUT     PORT3
1065 1227 11 00 30              LXI     D,TTDIG
1066 122A 1A                    LDAX    D       ;SAVE
1067 122B F5                    PUSH    PSW     ;IT
1068 122C 3A 00 30      PTCH6:  LDA     TTDIG
1069 122F B7                    ORA     A
1070 1230 CA 54 12              JZ      PTCH7
1071 1233 CD 74 01              CALL    DELAY
1072 1236 3D                    DCR     A
1073 1237 32 00 30              STA     TTDIG
1074 123A 13                    INX     D
1075 123B 1A                    LDAX    D       ;GET DIGIT
1076 123C 3D                    DCR     A
1077 123D 21 06 13              LXI     H,TTTAB
1078 1240 85                    ADD     L
1079 1241 6F                    MOV     L,A
1080 1242 7C                    MOV     A,H
1081 1243 CE 00                 ACI     0
1082 1245 67                    MOV     H,A
1083 1246 7E                    MOV     A,M     ;TTCODE
1084 1247 2F                    CMA
1085 1248 D3 70                 OUT     PORT7   ;SEND TONE
1086 124A CD 74 01              CALL    DELAY
1087 124D 3E FF                 MVI     A,0FFH
1088 124F D3 70                 OUT     PORT7   ;TONE OFF
1089 1251 C3 2C 12              JMP     PTCH6   ;NEXT
1090 1254 11 F8 30      PTCH7:  LXI     D,OUT3M
1091 1257 06 40                 MVI     B,40H
1092 1259 CD 10 02              CALL    BITC    ;LINE NORMAL
1093 125C 47                    MOV     B,A
1094 125D 3A F1 30              LDA     NOTIM
1095 1260 B0                    ORA     B
1096 1261 D3 30                 OUT     PORT3
1097 1263 F1                    POP     PSW
1098 1264 32 00 30              STA     TTDIG   ;RESTORE IT
1099 1267 C3 FC 01              JMP     TTON2   ;DONE
1100 126A 3A 01 30      PTCH8:  LDA     TTDIG+1
1101 126D FE 01                 CPI     1
1102 126F CA FC 01              JZ      TTON2
1103 1272 C3 02 12              JMP     PTCH3
1104 1275 21 3F 13      PTCH9:  LXI     H,SDTAB
1105 1278 3A 01 30              LDA     TTDIG+1
1106 127B 47                    MOV     B,A
1107 127C 7E            PCH10:  MOV     A,M
1108 127D B7                    ORA     A
1109 127E CA FC 01              JZ      TTON2
1110 1281 B8                    CMP     B
1111 1282 CA 8B 12              JZ      PCH11
1112 1285 23                    INX     H
1113 1286 23                    INX     H
1114 1287 23                    INX     H
1115 1288 C3 7C 12              JMP     PCH10
1116 128B 23            PCH11:  INX     H
1117 128C 5E                    MOV     E,M
1118 128D 23                    INX     H
1119 128E 56                    MOV     D,M
1120 128F EB                    XCHG
1121 1290 7E                    MOV     A,M
1122 1291 B7                    ORA     A
```

forward references from the lower ROM go to the beginning of the second ROM, which will not change if a routine in the second ROM is modified. Frequent use is made of reading an address from a fixed location rather than reading an address directly.

The code table is organized with a three-digit code preceding the address of the program to service that code. The end of the table is marked with a 0. Naturally, the published codes are not the ones in use. The CW ID messages are set up with leading and trailing spaces to clean up the ID.

The RAM has the bottom 25 bytes reserved for the digit buffer, including one for the buffer length. 12 bytes are reserved above that for the digit #1 telephone number. Above that, space is left for the programmable ID. 22 bytes at the top are variables, and

```
1123 1292 CA FC 01          JZ      TTON2
1124 1295 FE 0C             CPI     12
1125 1297 D2 FC 01          JNC     TTON2
1126 129A 11 00 30          LXI     D,TTDIG
1127 129D 46                MOV     B,M
1128 129E 04                INR     B
1129 129F 7E          PCH12: MOV    A,M
1130 12A0 12                STAX    D
1131 12A1 23                INX     H
1132 12A2 13                INX     D
1133 12A3 05                DCR     B
1134 12A4 C2 9F 12          JNZ     PCH12
1135 12A7 3E 20             MVI     A,20H   ;DISABLE
1136 12A9 32 F1 30          STA     MOTIM   ;TIMER
1137 12AC C3 02 12          JMP     PTCH3
1138 12AF             ;
1139 12AF             ;
1140 12AF             ;
1141 12AF             ;REMOTE BASE CONNECTS RPT TO PHONE LINE
1142 12AF             ;BUT DOES NOT SEIZE THE LINE
1143 12AF 11 F9 30    RBASE: LXI    D,OUT1M
1144 12B2 06 04             MVI     B,4
1145 12B4 C3 D6 11          JMP     PTCH1
1146 12B7             ;
1147 12B7             ;
1148 12B7             ;
1149 12B7 3A F8 30    TAP2:  LDA    OUTOM   ;TAPE ACCESS
1150 12BA B7                ORA     A       ;VIA CONTROL
1151 12BB FA FC 01          JM      TTON2   ;STATION
1152 12BE C3 D7 02          JMP     TAP1
1153 12C1             ;
1154 12C1             ;
1155 12C1             ;
1156 12C1             ;DIAL SENDS IN CW WHATEVER WAS LAST
1157 12C1             ;ENTERED VIA THE TTT ROUTINE
1158 12C1             ;OR THE AUTOPATCH
1159 12C1 3A F8 30    DIAL:  LDA    OUTOM   ;WHAT DID I DIAL?
1160 12C4 E6 10             ANI     10H     ;ENABLED?
1161 12C6 C2 FC 01          JNZ     TTON2   ;NO
1162 12C9 3A 00 30          LDA     TTDIG
1163 12CC FE 19             CPI     25      ;VALID?
1164 12CE D2 FC 01          JNC     TTON2   ;NO
1165 12D1 CD 90 02          CALL    WCD
1166 12D4 C3 FE 10          JMP     TTSTO
1167 12D7             ;
1168 12D7             ;
1169 12D7             ;
1170 12D7             ;LOCK PERMITS A LOCKOUT OF CONTROL
1171 12D7             ;AND DISABLING OF THE ROGER ROUTINE
1172 12D7             ;LOCK WAITS FOR 3 DIGITS:
1173 12D7             ;THE SECOND ELIMINATES (1) OR CLEARS (0)
1174 12D7             ;THE ROGER ROUTINE,
1175 12D7             ;THE THIRD LOCKS OR UNLOCKS ENTRY
1176 12D7             ;TO THE CONTROL MODE
1177 12D7 CD 19 02    LOCK:  CALL    LOAD
1178 12DA E 01              MVI     A,1
1179 12DC B8                CMP     E
1180 12DD C2 E3 12          JNZ     LOCK3
1181 12E0 C3 EA 12          JMP     LOCK4
1182 12E3 3E 0A       LOCK3: MVI    A,10
1183 12E5 B8                CMP     E
1184 12E6 C2 ED 12          JNZ     LOCK5
1185 12E9 AF                XRA     A
1186 12EA 32 F7 30    LOCK4: STA    LKROG
1187 12ED 3E 01       LOCK5: MVI    A,1
1188 12EF B8                CMP     B
1189 12F0 C2 F6 12          JNZ     LOCK1
1190 12F3 C3 FD 12          JMP     LOCK2
1191 12F6 3E 0A       LOCK1: MVI    A,10
1192 12F8 B8                CMP     B
1193 12F9 C2 FC 01          JNZ     TTON2   ;INVALID
1194 12FC AF                XRA     A
1195 12FD 32 F2 30    LOCK2: STA    LCKR
1196 1300 CD D2 03          CALL    ROGER
1197 1303 C3 FC 01          JMP     TTON2
1198 1306             ;
1199 1306             ;
1200 1306             ;
1201 1306             ;TABLE FOR REGENERATING TOUCH TONES (R)
1202 1306 88          TTTAB: DB     88H     ;1
1203 1307 84                DB      84H     ;2
1204 1308 82                DB      82H     ;3
1205 1309 48                DB      48H     ;4
1206 130A 44                DB      44H     ;5
1207 130B 42                DB      42H     ;6
1208 130C 28                DB      28H     ;7
1209 130D 24                DB      24H     ;8
1210 130E 22                DB      22H     ;9
1211 130F 14                DB      14H     ;0
1212 1310 18                DB      18H     ;*
1213 1311 12                DB      12H     ;#
1214 1312             ;
1215 1312             ;
1216 1312             ;
1217 1312             ;LOAD NUMBER FOR SINGLE DIGIT #1
1218 1312 21 19 30    LNUM:  LXI    H,NUMBR
1219 1315 36 00             MVI     M,0
1220 1317 DB 10       LNUM1: IN     PORT1
1221 1319 2F                CMA
1222 131A E6 40             ANI     40H
1223 131C CA 17 13          JZ      LNUM1
1224 131F CD 3B 02          CALL    DECOD
1225 1322 FE 0B             CPI     11      ;*
1226 1324 C2 2D 13          JNZ     LNUM2
1227 1327 CD D2 03          CALL    ROGER
1228 132A C3 FC 01          JMP     TTON2
1229 132D 47          LNUM2: MOV    B,A
1230 132E 3A 19 30          LDA     NUMBR
1231 1331 FE 0B             CPI     11      ;MAX DIGITS
1232 1333 CA 17 13          JZ      LNUM1
1233 1336 3C                INR     A
1234 1337 32 19 30          STA     NUMBR
1235 133A 23                INX     H
1236 133B 70                MOV     M,B
1237 133C C3 17 13          JMP     LNUM1
1238 133F             ;
1239 133F             ;
1240 133F             ;
1241 133F 01          SDTAB: DB     1
1242 1340 19 30             DW      NUMBR   ;#1
1243 1342 02                DB      2
1244 1343 58 13             DW      TNUM2
1245 1345 03                DB      3
1246 1346 63 13             DW      TNUM3
1247 1348 04                DB      4
1248 1349 6B 13             DW      TNUM4
1249 134B 05                DB      5
1250 134C 73 13             DW      TNUM5
1251 134E 06                DB      6
1252 134F 7B 13             DW      TNUM6
1253 1351 07                DB      7
1254 1352 83 13             DW      TNUM7
1255 1354 08                DB      8
1256 1355 8B 13             DW      TNUM8
1257 1357 09                DB      9
1258 1358 93 13             DW      TNUM9
1259 135A 00                DB      0
1260 135B             ;
1261 135B 07          TNUM2: DB     7       ;BALTO CITY
1262 135C 02                DB      2
1263 135D 02                DB      2
1264 135E 02                DB      2
1265 135F 03                DB      3
1266 1360 03                DB      3
1267 1361 03                DB      3
1268 1362 03                DB      3
1269 1363 07          TNUM3: DB     7       ;TRANSIT & TFC
1270 1364 03                DB      3
1271 1365 09                DB      9
1272 1366 06                DB      6
1273 1367 03                DB      3
1274 1368 0A                DB      10
1275 1369 05                DB      5
1276 136A 0A                DB      10
1277 136B 07          TNUM4: DB     7       ;MD STATE POL
1278 136C 04                DB      4
1279 136D 08                DB      8
1280 136E 06                DB      6
1281 136F 03                DB      3
1282 1370 01                DB      1
1283 1371 0A                DB      10
1284 1372 01                DB      1
1285 1373 07          TNUM5: DB     7       ;HARBOR TUNNEL
1286 1374 03                DB      3
1287 1375 05                DB      5
1288 1376 05                DB      5
1289 1377 03                DB      3
1290 1378 05                DB      5
1291 1379 0A                DB      10
1292 137A 0A                DB      10
1293 137B 07          TNUM6: DB     7       ;ANNE ARUNDEL
1294 137C 09                DB      9
1295 137D 08                DB      8
1296 137E 07                DB      7
1297 137F 04                DB      4
1298 1380 0A                DB      10
1299 1381 05                DB      5
1300 1382 0A                DB      10
1301 1383 07          TNUM7: DB     7       ;COAST GUARD
1302 1384 07                DB      7
1303 1385 08                DB      8
1304 1386 09                DB      9
1305 1387 08                DB      8
1306 1388 0A                DB      10
1307 1389 05                DB      5
1308 138A 0A                DB      10
1309 138B 07          TNUM8: DB     7       ;BALTO CO
1310 138C 04                DB      4
1311 138D 09                DB      9
1312 138E 04                DB      4
1313 138F 02                DB      2
1314 1390 01                DB      1
1315 1391 01                DB      1
1316 1392 01                DB      1
1317 1393 07          TNUM9: DB     7       ;HOWARD CO
1318 1394 04                DB      4
1319 1395 06                DB      6
1320 1396 05                DB      5
1321 1397 01                DB      1
1322 1398 06                DB      6
1323 1399 01                DB      1
1324 139A 01                DB      1
1325 1398             ;
1326 1398             ;
1327 1398             ;
1328 1398                   ORG     3000H   ;RAM BOTTOM
1329 3000             ;
1330 3000             TTDIG: EQU    $
1331 3000                   DS      25      ;SPACE FOR DIGITS
1332 3019             NUMBR: DS     12      ;TEL #1
1333 3025             IDADS: EQU    $
1334 3025                   DS      197     ;SPACE FOR STACK, ID #5
1335 30EA             STACK: EQU    $
1336 30EA             OUTR1: DS     1
1337 30EB             OUTR2: DS     1
1338 30EC             OUTR3: DS     1
1339 30ED             TIMER: DS     4
1340 30F1             MOTIM: DS     1
1341 30F2             LCKR:  DS     1
1342 30F3             IOS:   DS     1
1343 30F4             IOM:   DS     1
1344 30F5             TIME:  DS     1
1345 30F6             MASK:  DS     1
1346 30F7             LKROG: DS     1
1347 30F8             OUTOM: DS     1
1348 30F9             OUT1M: DS     1
1349 30FA             OUT2M: DS     1
1350 30FB             OUT3M: DS     1
1351 30FC             OUT4M: DS     1
1352 30FD             OUT5M: DS     1
1353 30FE             OUT6M: DS     1
1354 30FF             OUT7M: DS     1
```

the stack starts below them. The stack works down, and the programmable ID works up. No safeguards are set up to eliminate the two clashing. The amount of space is so large for the required functions that for even the longest imaginable ID message there will be plenty of room left for the stack. I do not suggest testing the system by loading an ID of 197 characters! Up to 150 should be safe. OUT0M is a dummy output port. Although it is set up as an output port, there is no physical port. This is convenient for both programming and operation.

Design Philosophy

As previously mentioned, several years ago I had constructed a microprocessor system to perform a similar

function. At that time, I built the hardware first. After completing this project, I have no doubt that the proper procedure is the other way around. A general idea of the hardware should be in mind, but the program should be written first. Writing the program defines the parameters of the system. By doing so, I found that some hardware modifications were needed that otherwise I would have had to go back and redo.

The program was written and debugged on the development system described. I configured the I/O ports so that the program could be executed on my large system. The program was in operation on it before a single wire was cut to construct the hardware. Clip leads and external oscillators were used to test the system. Did you ever try to simulate touchtones with clip leads, trying not to be caught by a three-second timer?

The program was modularized as much as possible. If any routine is longer than about two or three pages, it is too long and should be broken down into smaller routines. Not only is it easier to write that way, but it is also easier to understand how it works a few months later. For routines with many conditionals, flowcharts are a must. Originally, a skeleton program was written—just enough so that the entire program was self-consistent. Gradually the individual routines can be added to the code table and debugged. The throughput using these techniques can be quite high. I wrote the skeleton program in one day, and debugged it the next. Once an operational program was ready, the hardware was constructed. In the week or so it took to build the thing, the program was beefed up. By the time the hardware was ready, the software was refined. I cannot overemphasize the fact that a 100% operational program is necessary before building the hardware. When the ROMs are plugged in, if the program is in any doubt, and the system does not work, you do not know if the problem is hardware or software, resulting in an exercise in futility.

The hardware/software tradeoffs previously mentioned are important. A lot of thought is necessary before plunging ahead with design. The total software and hardware development time/cost must be considered. Even though the individual pays nothing for his own software, thinking like the businessman who must pay for his software will give a more balanced design.

When building hardware, it is advantageous to freely add LEDs on signal lines. You may not need them after the circuitry is in operation, but they are invaluable when debugging and testing the system. Design a system that not only works properly, but also can be made to operate properly in a reasonable amount of time.

Fault tolerance is an area at the frontier of theoretical knowledge. The discipline is about a decade old, and much remains to be worked out satisfactorily. Semiconductor technology is increasing at a rate which is hard to keep up with. Writing programs which merely function, and programs which both function and are error-tolerant, are two different things. Instead of making equivalence tests, it is better to make relational tests. Otherwise, if an error occurs, a test may fall through. Subprograms are usually expected to be entered with certain initial conditions. They should be constructed so that if those conditions are erroneous, the subprogram will exit soon. The worst thing that can happen is an erroneous input condition resulting in an endless loop. In a controller, it may not be as easy to push the reset button when something goes awry as it is on a general-purpose computer. I certainly did not follow all of these tenets in writing the software; however, I attempted to keep them in mind as much as possible.

The original program, somehow, did manage to crash twice. After that, I added the error recovery portion. It is a very simple, first-order attack, but it covers more errors than a first glance shows. If the program gets into a false state, it will often go to a faulty address. Since the hardware uses a small amount of the address space, it is quite likely that the program will be sent to



Fig. 8. ID load routine.



Fig. 9. Out routine.



Fig. 10. Load number and lock routines.



Fig. 11. Autopatch routine.

a place where there is no memory. This results in reading all highs; the instruction FF is the interrupt instruction, so effectively an invalid memory address interrupts the program. That is why I placed the recovery routine at the interrupt location. The processor is not being interrupted, but it interprets the error as an interrupt. A second different thing about the fault-tolerant program is that the enable interrupt instruction was placed into the master loop. Otherwise, if the interrupt were ever disabled when in the foreground program, there would be no way to communicate with it.

I am not claiming that the system is totally fault-tolerant, but by the addition of some very simple checks, the fault tolerance can be increased tremendously. This entire project has been a good education.

Expansion

There are many additions and improvements which can be made. The advantage of the whole arrangement is that for many changes, hardware need not be touched. Many functions can be added by software changes only. It is more pleasant to sit in an easy chair at home rewriting the program than to sit on the cold, hard floor at the repeater site to effect changes. If changes don't work, all that has to be done is to put the original ROMs back in.

Additional hardware can be added to mate with the existing circuitry, and it is not necessary to worry about the additional control functions, as plenty of spares are already provided. A possible improvement to the software would allow interrogation of the status bits. This is a simple addition which is not required but might be useful. A planned hardware addition to the system will provide downlink telemetry from the site. Lights on the voting selector indicate which receivers are being accessed, and which receiver the voter selects. The telemetry will transmit the voter lights in real-time. Incorporated in the telemetry package will be an analog-to-digital converter. Upon command from the control system, the telemetry will switch from the voter lights to meter readings read by the A/D; plate voltage, plate and grid current for each repeater, and cabinet temperature could be read. With the existing central control system, the possibilities for expansion are straightforward and exciting.

Acknowledgements

I would like to thank Carroll Van Ness K3HZU for his able assistance in designing the autopatch circuitry. Until this time, Carroll has been the father of the control circuitry and the autopatch. His equipment always functioned fine, but there is only so much that can be done with relays. Carroll is now a microcomputer convert.

I received help from Vern Chapin K3VC with the metalwork. Despite broken saw blades and bruised fingers, he finished the panels.

I thank Frank Ayd WA3ILR, who stayed with me over 13 hours at the site on the day of installation. We were both dirty, tired, cold, and hungry, but he remained with me while making frantic pleas that we quit.

Thanks go again to Vern Chapin K3VC, and also to Marc Leavey, M.D., WA3AJR, for their photography.

And if not for Jack Biggs K3SP and Larry D'Anna WA3KOK, with the assistance of many others over a period of several years, I would not have had the excellent repeater for which to develop the control system. ■